* ==============================================================================
*
* Date:	Summer 2024
* Paper: Climate shocks, regional favoritism and trust in leaders: Insights from droughts in Africa
*
* This creates Table S8
*
* ==============================================================================

* Open dataset
cd $PATH
use tr_rep, clear

* Global macros

* Control sets
global xcont "age age2 male i.ed_ord"

global fecont "urban ea_svc_a ea_svc_b ea_svc_c ea_fac_a ea_fac_b ea_fac_c ea_fac_d ea_fac_e ea_sec_a ea_sec_b grs0 grsem0 grsl cropland_d ln_alt ln_dist_cit ln_alt_sd ln_dist_bor ln_dist_coa ln_dist_riv ln_dist_roa i.year i.month i.uid1 countrytrend*"

* Clustering level
global clustervar "cidgrid01"

**#

* Clean sample
keep if precision_code<=3
global zxcont "age age2 male ed_ord"
global zfecont "urban ea_svc_a ea_svc_b ea_svc_c ea_fac_a ea_fac_b ea_fac_c ea_fac_d ea_fac_e ea_sec_a ea_sec_b grs0 grsem0 grsl cropland_d ln_alt ln_dist_cit ln_alt_sd ln_dist_bor ln_dist_coa ln_dist_riv ln_dist_roa year"

foreach var of varlist $zxcont $zfecont {
	drop if `var'==.
}

* Standardize
egen z=std(aiclwmcgrwsp)
replace aiclwmcgrwsp=z
drop z

* Table name
global tabname "baseline_poltrust"

* Dependent variables
global dvars "tr_pres_ord tr_parl_ord tr_lgov_ord tr_part_ord tr_opp_ord tr_trad_ord"

* Regression(s)
timer clear
local x=0
local z = ""

foreach dvar of varlist $dvars {
	
*c
local x=`x'+1
di "Column `x'"
timer on `x'
qui reg `dvar' aiclwmcgrwsp $xcont $fecont, vce(cluster $clustervar)
local z`x' m`x'
local z = "`z'" + " `z`x''"
est store m`x'
local m="c`x'"
estimates save $tabname`m', replace
timer off `x'
timer list `x'
}


* Post-table 
timer list   
di "`z'"

* Simplified table generation
global keepvars "aiclwmcgrwsp"

* In Stata output window
esttab `z' , ///
	mtitles("President" "Parliament" "Local gov." "Ruling party" "Opposition parties" "Traditional leaders") ///
	addnotes("Standard errors clustered on 0.1-degree grid cells") ///
	keep($keepvars) ///
	order($keepvars) ///
	cells(b(star fmt(%8.3f) ) se( par fmt(%6.3f)) ) ///
	stats(N r2, fmt(%9.0f %9.2f)) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) r2(3) ///
	legend


* Excel-file
estout `z' using $OUTPATH/$tabname.xls, replace ///
	mlabels("President" "Parliament" "Local gov." "Ruling party" "Opposition parties" "Traditional leaders") ///
	postfoot("Standard errors clustered on 0.1-degree grid cells") ///
	keep($keepvars) ///
	order($keepvars) ///
	cells(b(star fmt(%8.3f) ) se( par(`"="("' `")""') fmt(%6.3f)) ) ///
	stats(N r2, fmt(%9.0f %9.2f)) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) ///
	legend

